

(* Asymptotic expected number of hairpins in ALL structures with theta=1 and stickiness p=1/2*)

(*We first compute the dominant singularity and asymptotic number of saturated structures, using Drmota–Lalley–Woods Theorem. Grammar with S, T where T means there is at least one base pair. *)

```

Clear["*"]
(* S arrow D or T; T arrow (E) or T* or (T) or S(D1) or S(T); D0 arrow * or D0 *;
D1 arrow *** or D1 * *)
Clear[p, S, D0, D1, z, eqn0, eqn, F, z0, y0, dFdzoFz0S0, d2FdyOfz0S0];

eqn0 = {S == z + z S + z^2 S + z^2 S^2}; (* grammar for all str, theta=1,p=1 *)
CellPrint["eqn0 is the usual (simple) grammar for all sec str, with theta=1,p=1"]
eqn1 = {S == D0 + T, T == z^2 D1 + z T + z^2 T + z^2 S T + z^2 S D1,
        D0 == z + z D0, D1 == z + z D1};
CellPrint[
  "eqn1 is variant of eqn,where D1 arrow * or D1 *, giving equivalent grammar to eqn0"]
Eliminate[eqn0, {D0, D1, T}]
Collect[%, S, Simplify]
Eliminate[eqn1, {D0, D1, T}]
Collect[%, S, Simplify]
CellPrint["We derive the SAME equations
  from eqn0,eqn1 showing the underlying grammar is correct."]
p = 1 / 2;
eqn = {S == D0 + T, T == p z^2 D1 + z T + p z^2 T + p z^2 S T + p z^2 S D1,
        D0 == z + z D0, D1 == z + z D1};
Eliminate[eqn, {D0, D1, T}]
Collect[%, S, Simplify]
CellPrint["From previous expression, write S = ... to get F"]
F = (S^2 z^2 + 2 z) / - (-2 + 2 z + z^2);

NSolve[{F == S, D[F, S] == 1}, {z, S}]
z0 = 0.4575768754160631`;
y0 = 2.0906578508522444`;

dFdzoFz0S0 = D[F, z] /. {z -> z0, S -> y0}

d2FdyOfz0S0 = D[F, {S, 2}] /. {z -> z0, S -> y0}
c = Sqrt[z0 dFdzoFz0S0 / (2 Pi d2FdyOfz0S0)]
c * (1 / z0)^n n^(-3 / 2)

```

eqn0 is the usual (simple) grammar for all sec str, with theta=1,p=1

eqn1 is variant of eqn,where D1 arrow * or D1 *, giving equivalent grammar to eqn0

$$S^2 z^2 + S (-1 + z + z^2) = -z$$

$$S^2 z^2 + S (-1 + z + z^2) = -z$$

$$S^2 z^2 + S (-1 + z + z^2) = -z$$

$$S^2 z^2 + S (-1 + z + z^2) = -z$$

We derive the SAME equations from eqn0,eqn1 showing the underlying grammar is correct.

$$S^2 z^2 + S (-2 + 2z + z^2) = -2z$$

$$S^2 z^2 + S (-2 + 2z + z^2) = -2z$$

From previous expression, write $S = \dots$ to get F

```
{ {z → 4.37085, S → -0.676444}, {z → 0.457577, S → 2.09066},
  {z → -0.414214 + 1.35219 i, S → -0.707107 + 0.956145 i},
  {z → -0.414214 - 1.35219 i, S → -0.707107 - 0.956145 i}}
```

13.815

0.478318

1.4503

$$\frac{1.4503 \times 2.18543^n}{n^{3/2}}$$

(*Now, we compute mean,variance using Drmota's Theorem*)

```
Clear["*"]
p = 1 / 2;
eqn = {S == D0 + T, T == p u z^2 D1 + z T + p z^2 T + p z^2 S T + p u z^2 S D1,
  D0 == z + z D0, D1 == z + z D1};

CellPrint["Eliminate all variables except S,u,z"]
Eliminate[eqn, {D0, D1, T}]
Collect[%, S, Simplify]
CellPrint["From previous expression, write S=... to obtain F"]
F = (S^2 (-1 + z) z^2 + z (-2 + 2 z + z^2) - (u z^3 + S u z^3)) / - (2 - 4 z + z^2 + 2 z^3);

f = (F /. S → s)
s - f
(* express over a common denominator*)
Together[s - f]
a = Numerator[%]
c = Denominator[Together[s - f]]
dfs = D[f, s]
1 - dfs
(* express over a common denominator*)
Together[1 - dfs]
(* a is numerator of s-f, and b is numerator of 1-dfs,
where both have identical denominators *)
b = Numerator[%]
d = Denominator[Together[1 - dfs]]
(* a is numerator of s-f, and b is numerator of 1-dfs,
where both have identical denominators *)
If[Denominator[Together[s - f]] == Denominator[Together[1 - dfs]],
  CellPrint["Denominator of (s-f) same as that of (1-dfs)"],
  CellPrint["Denominator of (s-f) different than that of (1-dfs)"]]
CellPrint["Now compute the resultant of numerators a,b to get relation between S,u,z"]

(* res =Resultant[s-f,1-dfs,s] *)
```

```
(*We compute resultant of numerators,
since we have Resultant[S-F,1-D[F,S],S]=0 which is not much use.*)
CellPrint["Express S-F and 1-D[F,S] as fractions over the same common denominator"]
CellPrint["Then compute the resultant of the numerators of these expressions"]
res = Resultant[a, b, s]
(* Replace z by z[u], a function of u *)

res /. z -> z[u]
(* Now compute z'[u] *)
dres = D[%, u]
Simplify[Collect[dres, z'[u]]]
Solve[dres == 0, z'[u]]
dzu = Last[Last[Last[Solve[dres == 0, z'[u]]]]];
(*z[1] equals rho, the dominant singularity *)

rho = 0.4575768754160631`;
(* value of z0 in the first part of this file, the dominant singularity*)
dzuEvaluatedAt1 = (dzu /. u -> 1) /. z[1] -> rho
CellPrint[
  "According to Drmota's Theorem 1, the mean equals -z'[1]/z[1], computed next. "]

mu = ((-dzu / z[u] /. u -> 1) /. z[1] -> rho)

(*Variance computation *)
(* d2zu is z''[1] *)
d2zu = ((D[dzu, u] /. u -> 1) /. z[1] -> rho) /. z'[1] -> dzuEvaluatedAt1
CellPrint["Now compute variance, which by Drmota is -z''[1]/z[1] + mu^2 + mu"]
var = -d2zu / rho + mu * mu + mu
CellPrint["Now compute the standard deviation sigma"]
stdev = Sqrt[var]
```

Eliminate all variables except S,u,z

$$2 S - 2 z - 4 S z + 2 z^2 + S z^2 - S^2 z^2 + z^3 + 2 S z^3 + S^2 z^3 = (1 + S) u z^3$$

$$S^2 (-1 + z) z^2 + z (-2 + 2 z + z^2) + S (2 - 4 z + z^2 + 2 z^3) = u z^3 + S u z^3$$

From previous expression, write S=... to obtain F

$$s = \frac{s^2 (-1 + z) z^2 - u z^3 - s u z^3 + z (-2 + 2 z + z^2)}{-2 + 4 z - z^2 - 2 z^3}$$

$$s = \frac{s^2 (-1 + z) z^2 - u z^3 - s u z^3 + z (-2 + 2 z + z^2)}{-2 + 4 z - z^2 - 2 z^3}$$

$$\frac{2 s - 2 z - 4 s z + 2 z^2 + s z^2 - s^2 z^2 + z^3 + 2 s z^3 + s^2 z^3 - u z^3 - s u z^3}{2 - 4 z + z^2 + 2 z^3}$$

$$\frac{2 s - 2 z - 4 s z + 2 z^2 + s z^2 - s^2 z^2 + z^3 + 2 s z^3 + s^2 z^3 - u z^3 - s u z^3}{2 - 4 z + z^2 + 2 z^3}$$

$$\frac{2 s (-1 + z) z^2 - u z^3}{-2 + 4 z - z^2 - 2 z^3}$$

$$1 - \frac{2s(-1+z)z^2 - uz^3}{-2+4z-z^2-2z^3}$$

$$\frac{2-4z+z^2-2sz^2+2z^3+2sz^3-uz^3}{2-4z+z^2+2z^3}$$

$$2-4z+z^2-2sz^2+2z^3+2sz^3-uz^3$$

$$2-4z+z^2+2z^3$$

Denominator of (s-f) same as that of (1-dfs)

Now compute the resultant of numerators a,b to get relation between S,u,z

Express S-F and 1-D[F,S] as fractions over the same common denominator

Then compute the resultant of the numerators of these expressions

$$4z^2 - 20z^3 + 36z^4 - 28z^5 - 4uz^5 + 9z^6 + 12uz^6 - z^7 - 14uz^7 + 6uz^8 + u^2z^8 - u^2z^9$$

$$4z[u]^2 - 20z[u]^3 + 36z[u]^4 - 28z[u]^5 - 4uz[u]^5 +$$

$$9z[u]^6 + 12uz[u]^6 - z[u]^7 - 14uz[u]^7 + 6uz[u]^8 + u^2z[u]^8 - u^2z[u]^9$$

$$- 4z[u]^5 + 12z[u]^6 - 14z[u]^7 + 6z[u]^8 + 2uz[u]^8 - 2uz[u]^9 + 8z[u] \text{Derivative}[1][z][u] -$$

$$60z[u]^2 \text{Derivative}[1][z][u] + 144z[u]^3 \text{Derivative}[1][z][u] - 140z[u]^4 \text{Derivative}[1][z][u] -$$

$$20uz[u]^4 \text{Derivative}[1][z][u] + 54z[u]^5 \text{Derivative}[1][z][u] +$$

$$72uz[u]^5 \text{Derivative}[1][z][u] - 7z[u]^6 \text{Derivative}[1][z][u] - 98uz[u]^6 \text{Derivative}[1][z][u] +$$

$$48uz[u]^7 \text{Derivative}[1][z][u] + 8u^2z[u]^7 \text{Derivative}[1][z][u] - 9u^2z[u]^8 \text{Derivative}[1][z][u]$$

$$- 4z[u]^5 + 12z[u]^6 - 14z[u]^7 + 6z[u]^8 + 2uz[u]^8 - 2uz[u]^9 +$$

$$z[u] (8 - 60z[u] + 144z[u]^2 - 20(7+u)z[u]^3 + 18(3+4u)z[u]^4 -$$

$$7(1+14u)z[u]^5 + 8u(6+u)z[u]^6 - 9u^2z[u]^7) \text{Derivative}[1][z][u]$$

$$\left\{ \left\{ \text{Derivative}[1][z][u] \rightarrow - \left(2z[u]^4 - 6z[u]^5 + 7z[u]^6 - 3z[u]^7 - uz[u]^7 + uz[u]^8 \right) \right\} / \right.$$

$$\left. \left(-8 + 60z[u] - 144z[u]^2 + 140z[u]^3 + 20uz[u]^3 - 54z[u]^4 - \right. \right.$$

$$\left. \left. 72uz[u]^4 + 7z[u]^5 + 98uz[u]^5 - 48uz[u]^6 - 8u^2z[u]^6 + 9u^2z[u]^7 \right) \right\}$$

$$-0.045135$$

According to Drmota's Theorem 1, the mean equals $-z'[1]/z[1]$, computed next.

0.0986392

0.0352651

Now compute variance, which by Drmota is $-z''[1]/z[1] + \mu^2 + \mu$

0.0312998

Now compute the standard deviation sigma

0.176918